An Optimization Model for Maximizing Calories Burnt

Nutifafa Akpeleasi
Supervised by: Prof Melkonian Vardges

May 7, 2024

Abstract

With sedentary lifestyles contributing significantly to the rise of cardiovascular diseases, there's a growing need for effective exercise scheduling to maximize calorie expenditure within limited time frames. This study proposes an interval training exercise scheduling model using AMPL to optimize calorie burn during workouts. Motivated by previous research on circuit training optimization and interval running effects on caloric consumption, our model aims to select a combination of exercises and determine their durations to maximize calorie expenditure while considering constraints such as exercise intensity. We address the challenge of selecting exercises and determining their order to achieve optimal calorie burn. Our computational experiments demonstrate the effectiveness of the proposed model in selecting exercises and optimizing workout schedules to maximize calorie burn while ensuring safety and adherence to exercise guidelines.

Future work could explore incorporating constraints on athlete heart rate to further enhance the model's effectiveness and safety.

1 Introduction

Most people do exercise to achieve specific training goals. These goals may include weight reduction, reducing bad cholesterol levels in the blood and raise good cholesterol levels (the low-density lipoprotein (LDL) level), burning calories, improving muscular function and strength [1].

Sedentry lifestyles and physical inactivity are among the leading modifiable risk factors of cardiovascular diseases [2]. According to the Center for Health Disease and Stroke Prevention, cardiovascular disease is the leading cause of morbidity and mortality worldwide and so the need to provide a remedy that would promote physical activity and also help improve cardiovascular fitness is very important. Therefore, having an exercise schedule that maximizes the level of calories burnt given a set of exercises is crucial. In this current busy society, maintaining a good work-life balance has always been a challenge and as a result, making the best use of the little time we have in order to maximize our personal goals is significant.

There have been a number of work in providing a solution to this problem in the form of improving the physical activity of individuals and achieving specific training goals. In the case of circuit training, [3] built a linear integer model that minimizes the total circuit time while accomplishing a number of training goals. [4] also sought to determine how interval running activities affected acute caloric consumption. After an experiment with thirteen female and ten male sedentary collegian students, the results showed that interval running caused no more calorie expenditure than the classical steady state running acutely. [4] suggested the need for further work in this area.

In this project, we consider a variation of [3] and [4], where we build an interval training exercise scheduling model using A Mathematical Programming Language (AMPL) with the objective of maximizing the amount of calories burnt given a set of exercise in a given duration. According to [5] interval training is defined as alternating between brief periods of intense exercise and periods of less intense activity. This method can help individuals enhance their cardiovascular fitness, burn more calories, and attain better overall health and fitness results(This work is built on the work of Prof Melkonian Vardges [3]). In order to build this workout model to perform effectively, we consider some constraints and restrictions associated with the objective of maximizing calories.

- There is limited time available to complete the overall workout.
- An athlete may wish to alternate the intensity of the exercises to all some time for recovery.
- Each exercises should be completed within a specific timeframe since time is a constraint to an athlete or an individual.

2 The Model

With our basic model, our objective is to find the optimal level of calories that can be burnt given a set of workout routines. And so, we have a set of workouts and out of these, our model selects a number of them in order to burn as much calories as possible. We consider an interval training program.
In our model, we consider :

2.1 Sets

We have these two sets of data for our model.

$$W = \text{set of weights of the athlete} \tag{1}$$

$$Cal[i,j] = \text{calories an athlete with weight } j \text{ burns per minute doing exercise } i \tag{2}$$

2.2 Variables

$$X_i = \begin{cases} 1 & \text{if exercise } i \text{ is in the workout} \\ 0 & \text{if if not} \end{cases} \tag{3}$$

$$d_i = \text{duration of exercise i.} \tag{4}$$

$$f_{ij} = \begin{cases} 1 & \text{if exercise } j \text{ follows exercise } i \text{ in the workout} \\ 0 & \text{if if not} \end{cases} \tag{5}$$

$$EO_i = \text{the order of exercise } i \text{ in the workout} \tag{6}$$

2.3 Objective function

Our model is built to maximize the amount of calories burnt given a set of exercises and time duration. The model selects N exercises from a set of exercise that needs to be added to the workout and also determines how long each of these exercises should be performed:

$$max \sum_{i \in E} d_i * Cal[i,j]. \tag{7}$$

2.4 Constraints

Our model comes with some constraints which we discuss below.

(C1) The duration of each exercise selected for the workout must lie within a specified interval. Since we will have a restriction on the total time for the whole workout, it is prudent to restrict the duration of each exercise. If we do not have this constraint, the model may just pick the exercise with the highest calorie burn per minute and multiply it with the total duration with will defeat the objective of the interval training method. The maximum and minimum duration for each exercise is denoted by max_D and min_D respectively. The second constraint also ensures that if an exercise is not selected for the workout, then it must have no duration assigned to it. This makes sense because an athlete is not going to spend time completing an exercise that the model does not include in the workout.

$$min_D * X_i \leq d_i$$
and
$$d_i \leq max_D * X_i. \tag{8}$$

A detailed explanation and proof of the following constraints can be found at [3] where they were considered in building a circuit training model.

(C2) Our interval training model is similar to a graph with N nodes and $N-1$ edges. The model is going to select N exercises and the number of intervals between these exercises is $N-1$. $f_{ij} = 1$ if exercise j follows exercise i in the workout and 0 if not. The following constraint ensures that.

$$\sum_{i \in E, j \in E : i \neq j} f_{ij} = N - 1. \tag{9}$$

(C3) We want to make sure that exercise j will follow exercise i if both exercise i and j are included in the workout. That is:

$$f_{ij} \leq X_i \text{ and } f_{ij} \leq X_j. \tag{10}$$

(C4) The following constraint was adapted to ensure that we do not have a cycle in our training model [3]. We want to ensure that if an exercise selected for the workout, it does not get reselected. That is, our model should not suggest a particular exercise twice.

$$EO_i - EO_j + (N+1) \cdot f_{ij} \leq N, \tag{11}$$

where N is the number of exercises selected for the workout.

(C5) It makes sense to have our workout out in the form of a single chain and so we introduce a constraint to ensure that if exercise i is selected, then there is exactly on exercise preceding and succeeding. An athlete cannot do two exercises (example, aerobics and cycling) simultaneously and so this constraint prevents our model from giving the same rank to two or more exercises.

$$\sum_{j \in E : i \neq j} f_{ij} \leq 1 \tag{12}$$

$$\sum_{j \in E : i \neq j} f_{ji} \leq 1. \tag{13}$$

(C6) In order to ensure that our variable EO_i correctly ranks the selected exercises, we introduce the following constraints.

The rank of each exercise must lie between 0 and N with $EO_i = 0$ meaning that exercise i is not included in the workout and so it is not ranked and this is ensured by the following:

$$EO_i \leq N \left(\sum_{j \in E : j \neq i} (f_{ij} + f_{ji}) \right). \tag{14}$$

If exercise i is not selected for the workout, then $f_{ij} = f_{ji} = 0$, forcing $EO_i = 0$. Otherwise, exercise i will be ranked from 1 and N.

If we are selecting N exercises for the work out and we expect that each X_i will be assigned exactly one EO_i for $i \in \{1, 2, 3, \ldots N\}$. That is, for each i, X_i should **not** be assigned more than one rank. Hence $1 + 2 + 3 + \cdots + N - 1 + N = \dfrac{N(N+1)}{2}$. That is:

$$\sum_{i \in E} EO_i = \frac{N(N+1)}{2}. \tag{15}$$

(C7) With interval training, we want to alternate the intensity of the exercises that an athlete performs. One particular requirement for our model is to ensure that our model does not suggest a high intensity exercise directly after a high intensity exercise. That is, a high intensity exercise should be followed by a low intensity exercise. Our model maximizes the amount of calories burnt from the workout and calorie expenditure depends on the intensity of the workout. A way to measure the intensity of an exercise is given by the MET (*Metabolic Equivalent of Task*) values. According to National Institute of health [6], one metabolic equivalent (MET) is defined as the amount of oxygen consumed while sitting at rest and is equal to 3.5 ml O2 per kg body weight x min. [7] recommends a threshold MET value of 6 to distinguish between a low and high intensity exercise. The following constraints ensures that no two high intensity exercises share a direct path: For any two exercise

$$\begin{gathered} i \in E, j \in E : i \neq j, \\ f_{ij} = 0. \end{gathered} \tag{16}$$

2.5 The Complete Model

The following is the summary of our model:

$$max \sum_{i \in E} d_i * Cal[i, j] \quad (7)$$

subject to :

$$
\begin{aligned}
& min_D * X_i \leq d_i \\
& d_i \leq max_D * X_i && (8) \\
& f_{ij} \leq X_i \text{ and } f_{ij} \leq X_j && (10) \\
& EO_i - EO_j + (N+1) \cdot f_{ij} \leq N && (11) \\
& \sum_{j \in E: i \neq j} f_{ij} \leq 1 && (12) \\
& \sum_{j \in E: i \neq j} f_{ji} \leq 1 && (12) \\
& EO_i \leq N \left(\sum_{j \in E: j \neq i} (f_{ij} + f_{ji}) \right) && (14) \\
& \sum_{i \in E} EO_i = \frac{N(N+1)}{2} && (15) \\
& i \in E, j \in E : i \neq j, \\
& f_{ij} = 0. && (16)
\end{aligned}
$$

In the next section, we test model with real world data to see how it performs.

3 Computational Results

We convert our model into an AMPL code and we give the full model of the integer problem developed in the previous section.

3.1 AMPL Model

Listing 1: AMPL Code

```
# Example AMPL model
set Exercises; #The set of Exercises
set weights; # The weight categories of an athlete


param number_of_exercises;# number of exercises in a circuit
param calories{Exercises,weights}; #Calories burnt per minute for
    each exercise
param max_duration; #Maximum duration an exercise can be
   performed
param min_duration; #Minimum duration an exercise can be
   performed


##INTENSITY##
param MET{i in Exercises}; #The MET values of the exercises
param T_MET; #Threshold for distinguising between high intensity
   and low intensity exercises
param highMET{i in Exercises}:= if MET[i] >= T_MET then 1 else 0;
    #1 for high intensity and o for low intensity
param intensity_lower_limit;# Intensity lower limit
param intensity_upper_limit;# Intensity upper limit
param Time_Duration; #Duration of the interval training workout


##VARIABLES
var duration{Exercises} integer >=0; # duration of each exercise
var included{Exercises} binary; # This is 1 if exercise i is
   included in the selected workout and 0 otherwise.
var next_exercise{i in Exercises, j in Exercises: i!=j} binary;
   #1 if exercise i is folloed by exercise j and 0 otherwise.
var exercise_order{Exercises}>=0, <=number_of_exercises integer;#
    This determines the order of each exercise that is included
```

```
   in the workout.


maximize total_calories {j in weights}:
        sum{i in Exercises} duration[i]*calories[i,j];

##CONSTRAINTS##
subject to duration_within_this_interval_min {i in Exercises}: #
   duration of exercise i should be more than than minimum
   duration for each exercise
min_duration*included[i] <= duration[i] ;

subject to duration_within_this_interval_max {i in Exercises}: #
   duration of exercise i should be less than than maximum
   duration for each exercise
duration[i] <= max_duration * included[i];

subject to number_of_exercises_selected:
        sum{i in Exercises} included[i]=number_of_exercises; #
           Total number of exercises selected should be N

subject to total_duration:
        sum {i in Exercises} duration[i]<=Time_Duration;# The
           total duration of the selected models should not
           exceed the duration of the workout.

subject to number_of_following_exercises_in_training:
        sum{i in Exercises, j in Exercises: i!=j} next_exercise[i
           ,j] = number_of_exercises - 1;

subject to follow_exercise_only_if_included1{i in Exercises, j in
    Exercises: i!=j}:
        next_exercise[i,j] <= included[i];
# exercise i can be followed by exercise j only if exercise i is
   in the schedule

subject to follow_exercise_only_if_included2{i in Exercises, j in
    Exercises: i!=j}:
next_exercise[i,j] <= included[j];
### exercise i can be followed by exercise j only if exercise j
   is in the schedule
```

```
subject to at_most_one_exercise_follows{i in Exercises}:
sum{j in Exercises: i!=j} next_exercise[i,j] <= 1;

subject to at_most_one_exercise_precedes{j in Exercises}:
sum{i in Exercises: i!=j} next_exercise[i,j] <= 1;

subject to no_cycle{i in Exercises, j in Exercises: i!=j}:
exercise_order[i] - exercise_order[j] + (number_of_exercises + 1)
    * next_exercise[i,j] <= number_of_exercises;

subject to dummy_zero_if_not_included {i in Exercises}:
exercise_order[i] <= number_of_exercises * sum{j in Exercises: j
   !=i}(next_exercise[i,j] + next_exercise[j,i]);

subject to sum_of_orders:
sum{i in Exercises}exercise_order[i] = number_of_exercises * (
   number_of_exercises + 1)/2;

subject to no_two_high_intensities_in_a_row #A high intensity
   exercise should not follow a high intensity exercise and same
   applies to low intensity exercises
{i in Exercises, j in Exercises: i!=j and highMET[i]=1 and
   highMET[j]=1}:
next_exercise[i,j] = 0;
```

3.2 Input Data

We discuss our input data that we used to test out our model. As described earlier, the athlete has a set of exercises from which he or she has to choose N exercises, with the objective of maximizing the amount of calories burnt subject to some constraints.

List of Exercises

We test our model with 10 exercises. These exercises were selected based on three weight categories (130 lbs, 155 lbs and 180 lbs). The athlete can chose from more than 100 set of exercises with these weight categories that can be found at [8]. The list of exercises for our test data are: Aerobics (high impact), Aerobic (low impact), calisthenics (light), calisthenics (fast), cycling (> 20 mph), cycling (< 10 mph), jumping robe (fast), jumping robe (slow), strength hatha yoga, walking (3.0 mph).

Exercises included in the Interval Training

Based on the time availability of the individual, he or she may wish to vary the number of exercises in the interval training. To test our model, we require our model to select 6 exercises from the set of 10 exercises.

Duration of each exercise

In our model, we set the interval training time to 60 minutes and require that all the 6 exercises are performed within this time interval. And so our model will specify the time duration for each of the 6 exercises in the interval training in such a way that the calories burnt will be maximized.

Intensity values

The intensity of an exercise is determined by its MET value (Metabolic Equivalent of Task). According to [9], One metabolic equivalent (MET), which is equivalent to 3.5 ml O2 per kg body weight x min, is the quantity of oxygen used while sitting at rest. Our list of exercises has MET values between 3 and 16. A threshold value of 7 MET is used to distinguish between a high intensity exercise and a low intensity exercise. This threshold is recommended by [10]. Our list of exercises is made up of 5 low intensity exercises and 5 low intensity exercises. The goal is to alternate between high and low intensity levels.

Calorie values

The calories burnt for each of the 10 exercises can be found at [8]. The calories are based on the three weight categories (130 lbs, 155 lbs and 180 lbs). In testing our model, we rescaled the calories values to calories burnt per minute in order enable our model to calculate the calorie expenditure for any number of minutes.

3.3 Model Output for the Test Data

We fed our AMPL model with the test data as describes in the previous section and ran it using the CPLEX solver since we are dealing with a linear integer problem. Out of the 10 exercises from our test data, the model returned the following 6 exercises (weight less than 130 lb):

The total calories burnt displayed by our model for a weight less than 130 lb, less than 155 lb and less than 180 lb are 631.5 cal, 712.5 cal and 808 cal respectively.

Table 1: Output

Exercises selected	order performed	duration
Walking 3.0mph	1	5
Jogging general	2	15
Bicycling less than 10mph leisure	3	5
Bicycling more than 20mph racing	4	15
Aerobics general	5	5
Jumping rope fast	6	15

4 Future Work and Conclusion

In conclusion, our work presents an approach to interval training exercise scheduling using mathematical programming techniques. By optimizing the selection of exercises and their durations, our model aims to maximize calorie expenditure within a given time frame while considering constraints such as exercise intensity and time availability. Through computational experiments, we demonstrate the effectiveness of our model in designing workout schedules that maximize calorie burn while adhering to safety guidelines.

In future work, one can consider constraints on the athlete's target heart rate during each exercise. This addition would enhance safety by preventing the athlete's heart rate from surpassing recommended thresholds, thereby mitigating the risk of injuries or adverse health effects. By considering the impact of exercise intensity on heart rate, our model could provide more personalized and health-conscious workout schedules, catering to individual fitness levels and goals.

References

[1] J. Myers, "Exercise and cardiovascular health," *Circulation*, vol. 107, no. 1, pp. e2–e5, 2003.

[2] C. J. Lavie, C. Ozemek, S. Carbone, P. T. Katzmarzyk, and S. N. Blair, "Sedentary behavior, exercise, and cardiovascular health," *Circulation research*, vol. 124, no. 5, pp. 799–815, 2019.

[3] V. Melkonian, "An optimization model for exercise scheduling," *American Journal of Operations Research*, vol. 9, pp. 1–14, 2019.

[4] M. Yıldız and Z. Akyıldız, "The effects of interval running exercise on acute calorie expenditure," *Journal of Health Science*, vol. 5, pp. 360–365, 2017.

[5] M. M. Atakan, Y. Li, Ş. N. Koşar, H. H. Turnagöl, and X. Yan, "Evidence-based effects of high-intensity interval training on exercise capacity and health: A review with historical perspective," *International journal of environmental research and public health*, vol. 18, no. 13, p. 7201, 2021.

[6] M. Jetté, K. Sidney, and G. Blümchen, "Metabolic equivalents (mets) in exercise testing, exercise prescription, and evaluation of functional capacity," *Clinical cardiology*, vol. 13, no. 8, pp. 555–565, 1990.

[7] J. M. Hootman, "2008 physical activity guidelines for americans: an opportunity for athletic trainers," *Journal of athletic training*, vol. 44, no. 1, pp. 5–6, 2009.

[8] "Calories burned during exercise, activities, sports and work," https://www.nutristrategy.com/caloriesburned.htm, Accessed: April 20, 2024.

[9] B. Ainsworth, W. Haskell, S. Herrmann, N. Meckes, D. Bassett, C. Tudor-Locke, J. Greer, J. Vezina, M. Whitt-Glover, and A. Leon, "Compendium of physical activities: classification of energy costs of human physical activities," *Medicine and science in sports and exercise*, vol. 43, no. 8, pp. 1575–1581, 2011.

[10] M. d. A. Mendes, I. da Silva, V. Ramires, F. Reichert, R. Martins, R. Ferreira, and E. Tomasi, "Metabolic equivalent of task (mets) thresholds as an indicator of physical activity intensity," *PLoS One*, vol. 13, no. 7, p. e0200701, 2018.